Advanced Natural Study Guide, 4th Edition

Program Examples and Workshop Solutions


Chapter 2 – Map Editor
0010 ** EXCPGM01

0020 DEFINE DATA LOCAL

0030 1 EMPLOYEES VIEW OF EMPLOYEES

0040   2 PERSONNEL-ID

0050   2 NAME

0060   2 FIRST-NAME

0070   2 JOB-TITLE

0080 1 #PERSONNEL-ID(A8)

0090 END-DEFINE

0100 INPUT 2/30 'Vehicle Tracking System'

0110   4/20 'Enter Personnel ID:' #PERSONNEL-ID

0120 FIND EMPLOYEES WITH PERSONNEL-ID = #PERSONNEL-ID

0130   INPUT USING MAP 'EXCMAP01'

0140 END-FIND

0150 END

0010 ** EXCTEST1

0020 DEFINE DATA LOCAL

0030 1 EMPLOYEES VIEW OF EMPLOYEES

0040   2 PERSONNEL-ID

0050   2 NAME

0060   2 FIRST-NAME

0070   2 JOB-TITLE

0080 1 #PERSONNEL-ID(A8)

0090 END-DEFINE

0100 INPUT (AD='_') 2/30 'Vehicle Tracking System'

0110   4/20 'Enter Personnel ID:' #PERSONNEL-ID

0120 FIND EMPLOYEES WITH PERSONNEL-ID = #PERSONNEL-ID

0130   INPUT (IP=OFF SG=OFF ZP=OFF AD=MI'_')

0140     *DATX (EM=N(9)¬ZD,¬L(9))

0150     30T 'VEHICLE TRACKING SYSTEM'

0160     70T *TIMX (EM=HH:II¬AP)

0170     2/1 *INIT-USER

0180     2/30 'EMPLOYEE IDENTIFICATION'

0190     2/70 *PROGRAM

0200     3/1 '-' (79)

0210     5/30 'ENTER PERONNEL-ID:' EMPLOYEES.PERSONNEL-ID

0220     6/30 'ENTER LAST NAME:' EMPLOYEES.NAME

0230     7/30 'ENTER FIRST NAME:' EMPLOYEES.FIRST-NAME

0240     8/30 'ENTER JOB TITLE:' EMPLOYEES.JOB-TITLE

0250 END-FIND

0260 END

*

* EXCPGM01

*

DEFINE DATA

LOCAL

1 EMPLOYEES VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 FIRST-NAME

  2 NAME

  2 MIDDLE-I

  2 SEX

  2 BIRTH

  2 ADDRESS-LINE(1)

  2 CITY

  2 COUNTRY

  2 POST-CODE

  2 DEPT

  2 JOB-TITLE

  2 SALARY(1)

1 #PROGRAM(A8) INIT <*PROGRAM>

END-DEFINE

SET KEY ALL

SET KEY PF1 NAMED 'Help'

PF3 NAMED 'Exit'

READ (5) EMPLOYEES BY PERSONNEL-ID STARTING FROM '20000100'

  INPUT USING MAP 'EXCMAP02'

  IF *PF-KEY = 'PF3'

    ESCAPE BOTTOM

  END-IF

  IF *PF-KEY = 'PF1'

    REINPUT 'Help unavailable, please contact the help desk. '

  END-IF

END-READ

END

*

* EXCPGM02

*

DEFINE DATA LOCAL

1 EMPLOYEES VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 FIRST-NAME

  2 NAME

  2 MIDDLE-NAME

  2 SEX

  2 BIRTH

  2 ADDRESS-LINE(1)

  2 CITY

  2 COUNTRY

  2 ZIP

  2 DEPT

  2 JOB-TITLE

  2 SALARY(1)

1 #PROGRAM(A8) INIT <*PROGRAM>

1 #PERSONNEL-ID(A8)

END-DEFINE

SET KEY ALL

SET KEY PF1 NAMED 'Help'

PF3 NAMED 'Exit'

RPT01. REPEAT

  INPUT USING MAP 'EXCMAP00'

  READ (5) EMPLOYEES BY #PERSONNEL-ID

*     STARTING FROM '20000100'

    INPUT USING MAP 'EXCMAP01'

    IF *PF-KEY = 'PF3'

      ESCAPE BOTTOM (RPT01.)

    END-IF

    IF *PF-KEY = ‘PF1

      REINPUT 'Help unavailable, contact the help desk. '

    END-IF

  END-READ

END-REPEAT

END

Chapter 3 – Map Templates

0010 ** EXCPGM01

0020 DEFINE DATA LOCAL

0030 1 EMPLOYEES VIEW OF EMPLOYEES

0040   2 PERSONNEL-ID

0050   2 NAME

0060   2 FIRST-NAME

0070   2 MIDDLE-I

0080   2 SEX

0090   2 BIRTH

0100   2 ADDRESS-LINE(1:8)

0110   2 CITY

0120   2 COUNTRY

0130   2 POST-CODE

0140   2 DEPT

0150   2 JOB-TITLE

0160   2 SALARY(1:40)

0170 1 #PERSONNEL-ID(A8)

0180 1 #PROGRAM(A8) INIT <*PROGRAM>

0190 1 #SUBTITLE(A25) INIT <'  Employee Information'>

0200 END-DEFINE

0210 SET KEY ALL

0220 SET KEY PF1 NAMED 'Help'

0230   PF3 NAMED 'Exit'

0240 RPT1. REPEAT

0250   INPUT USING MAP 'EXCMAP00'

0260   IF *PF-KEY = 'PF3'

0270     ESCAPE BOTTOM (RPT1.)

0280   END-IF

0290   READ (5) EMPLOYEES BY PERSONNEL-ID STARTING FROM #PERSONNEL-ID

0300     INPUT USING MAP 'EXCMAP01'

0310     IF *PF-KEY = 'PF3'

0320       ESCAPE BOTTOM /* exits READ loop

0330     END-IF

0340     IF *PF-KEY = 'PF1'

0350       REINPUT 'Help unavailable; please contact the help desk.'

0360     END-IF

0370   END-READ

0380 END-REPEAT

0390 END

Chapter 4 – Arrays

0010 DEFINE DATA LOCAL

0020 1 #E48(A5/48)

0030 1 #I(I2)

0040 END-DEFINE

0050 FOR #I 1 48

0060 COMPRESS 'E-' #I INTO #E48(#I) LEAVING NO SPACE

0070 END-FOR

0080 INPUT USING MAP 'EXCMAPXX'

0090 END

0010 *

0020 * Program: EXCPGMX1 - Process 1000 element array example

0030 *

0040 DEFINE DATA LOCAL

0050 1 #ELEMENT-VALUES(A10/1000)

0060 1 #FROM(I4) INIT <1>

0070 1 #I(I4)

0080 END-DEFINE

0090 *

0100 * Load Array Values

0110 *

0120 FOR #I 1 1000

0130   COMPRESS 'Array-' #I INTO #ELEMENT-VALUES(#I) 

0140     LEAVING NO SPACE

0150 END-FOR

0160 *

0170 * Display Array Values in blocks of 50

0180 *

0190 FOR #I 1 20

0200   INPUT USING MAP 'EXCMAPXX' <-- name of map you generated

0210   ADD 50 TO #FROM

0220 END-FOR

0230 END

0010 ** EXCPGM02

0020 DEFINE DATA LOCAL

0030 1 EMPLOYEES VIEW OF EMPLOYEES

0040   2 PERSONNEL-ID

0050   2 NAME

0060   2 FIRST-NAME

0070   2 MIDDLE-I

0080   2 SEX

0090   2 BIRTH

0100   2 ADDRESS-LINE(1:8)

0110   2 CITY

0120   2 COUNTRY

0130   2 POST-CODE

0140   2 DEPT

0150   2 JOB-TITLE

0160   2 SALARY(1:40)

0170   2 BONUS (1:40,1:12) <---

0180 1 #PERSONNEL-ID(A8)

0190 1 #PROGRAM(A8) INIT <*PROGRAM>

0200 1 #SUBTITLE(A25) INIT <'  Employee Information'>

0210 END-DEFINE

0220 SET KEY ALL

0230 SET KEY PF1 NAMED 'Help'

0240   PF3 NAMED 'Exit'

0250 RPT1. REPEAT

0260   INPUT USING MAP 'EXCMAP00'

0270   IF *PF-KEY = 'PF3'

0280     ESCAPE BOTTOM (RPT1.)

0290   END-IF

0300   READ (5) EMPLOYEES BY PERSONNEL-ID 

0310       STARTING FROM #PERSONNEL-ID

0320     INPUT USING MAP 'EXCMAP01'

0330     IF *PF-KEY = 'PF3'

0340       ESCAPE BOTTOM /* exits READ loop

0350     END-IF

0360     IF *PF-KEY = 'PF1'

0370       REINPUT 'Help unavailable; please contact the help desk.'

0380     END-IF

0390   END-READ

0400 END-REPEAT

0410 END

0010 *

0020 * EXCPGM3D - 3D MAP EXAMPLE

0030 *

0040 DEFINE DATA LOCAL

0050 1 #ARRAY5X5X3(A10/5,5,3)

0060 1 #I(I4)

0070 1 #J(I4)

0080 1 #K(I4)

0090 END-DEFINE

0100 FOR #I 1 5

0110   FOR #J 1 5

0120     FOR #K 1 3

0130       COMPRESS #I '-' #J '-' #K INTO #ARRAY5X5X3(#I,#J,#K)

0140     END-FOR

0150   END-FOR

0160 END-FOR

0170 INPUT USING MAP 'EXCMAP3D'

0180 END

*

* EXCPGM03

*

DEFINE DATA

LOCAL

1 EMPLOYEES VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 FIRST-NAME

  2 NAME

  2 MIDDLE-NAME

  2 SEX

  2 BIRTH

  2 ADDRESS-LINE(1:2)

  2 CITY

  2 COUNTRY

  2 ZIP

  2 DEPT

  2 JOB-TITLE

  2 CURR-CODE(1:6)

  2 SALARY(1:6)

  2 BONUS(1:6,1:6)

1 #PROGRAM(A8) INIT <*PROGRAM>

END-DEFINE

SET KEY ALL

SET KEY PF1 NAMED 'Help'

  PF3 NAMED 'Exit'

READ (5) EMPLOYEES BY PERSONNEL-ID

    STARTING FROM '20000100'

  INPUT USING MAP 'EXCMAP01'

  IF *PF-KEY = 'PF3'

    ESCAPE BOTTOM

  END-IF

  IF *PF-KEY = 'PF'

    REINPUT 'Help is unavailable, please contact the help desk.'

  END-IF

END-READ

END

*

* LARGE ARRAYS, SNAKING COLUMNS, AND STARTING VARIABLES

* PROGRAM EXCP099A

*

DEFINE DATA LOCAL

203 379 :M

1 JOBS-VIEW VIEW OF EMPLOYEES

  2 JOB-TITLE

1 #JOBS(A25/240)

1 #BUZZ(I2)      /* ANYONE REMEMBER BUZZ?

1 #JOBS-IN-QUEUE(I2)

END-DEFINE

HISTOGRAM JOBS-VIEW JOB-TITLE

  MOVE *COUNTER TO #BUZZ

  MOVE JOB-TITLE TO #JOBS(#BUZZ)

  AT END OF DATA

    MOVE *COUNTER TO #JOBS-IN-QUEUE

  END-ENDDATA

END-HISTOGRAM

MOVE 1 TO #BUZZ

REPEAT UNTIL #BUZZ GT 200

  INPUT USING MAP ‘EXCMAP99

  ADD 60 TO #BUZZ

END-REPEAT

END

0010 *

0020 * Program: EXCPGMX1 - Process 1000 element array example

0030 *

0040 DEFINE DATA LOCAL

0050 1 #ELEMENT-VALUES(A10/1000)

0060 1 #FROM(I4) INIT <1>

0070 1 #I(I4)

0080 END-DEFINE

0090 SET KEY ALL

0100 SET KEY PF3 NAMED 'Exit'

0110   PF7 NAMED 'Back'

0120   PF8 NAMED 'Frwd'

0130 *

0140 * Load Array Values

0150 *

0160 FOR #I 1 1000

0170   COMPRESS 'Array-' #I INTO #ELEMENT-VALUES(#I) LEAVING NO SPACE

0180 END-FOR

0190 *

0200 * Display Array Values in blocks of 50

0210 *

0220 REPEAT

0230   INPUT USING MAP 'EXCMAPXX'

0240   IF *PF-KEY = 'PF3'

0250     ESCAPE BOTTOM

0260   END-IF

0270   IF *PF-KEY = 'PF8'

0280       OR = 'ENTR'

0290     ADD 50 TO #FROM

0300     IF #FROM GT 950

0310       MOVE 951 TO #FROM

0320     END-IF

0330   END-IF

0340   IF *PF-KEY = 'PF7'

0350     SUBTRACT 50 FROM #FROM

0360     IF #FROM LT 1

0370       MOVE 1 TO #FROM

0380     END-IF

0390   END-IF

0400 END-REPEAT

0410 END

Chapter 5 – Processing Rules


IF & = '20000000' THRU '29999999'


  IGNORE


ELSE REINPUT 'ID must begin with "2", please re-enter. '


  MARK FIELD *&


END-IF


DEFINE DATA LOCAL


1 VER-ID VIEW OF EMPLOYEES


  2 PERSONNEL-ID


END-DEFINE


H1. HISTOGRAM VER-ID PERSONNEL-ID


  IF *NUMBER(H1.) = 0


    REINPUT


  'There is no valid ID on file for:1: please re-enter.',&


      MARK FIELD *& (AD=B CD=RE)


  END-IF


END-HISTOGRAM


SET CONTROL 'WFC6L25B#'


INPUT WITH TEXT ' *** Personnel ID ***'


  'There is no ID on file'


  / 'for' & (AD=MI'. ')


  / 'please re-enter. '


SET CONTROL 'WB'

      END

0010 DEFINE DATA LOCAL                                        

0020 1 VER-ID VIEW OF EMPLOYEES                               

0030   2 PERSONNEL-ID                                         

0040 END-DEFINE                                               

0050 DEFINE WINDOW NO-ID                                      

0060   TITLE '*** Personnel ID ***'                           

0070   CONTROL SCREEN                                         

0080   FRAMED ON POSITION OFF                                 

0090 H01. HISTOGRAM VER-ID FOR PERSONNEL-ID FROM & THRU &     

0100 END-HISTOGRAM                                            

0110 IF *NUMBER(H01.) = 0                                     

0120   INPUT WINDOW='NO-ID'                                   

0130     'There is no ID on file'                             

0140     / 'for' & (AD=MI'_')                                 

0150     / 'please re-enter.'                                 

0160 END-IF

*

* SOLUTION 5.1.A - VERIFY PERSONNEL ID ON DATABASE

*

DEFINE DATA LOCAL

1 VALIDATE-ID VIEW OF EMPLOYEES

  2 PERSONNEL-ID

END-DEFINE

H1. HISTOGRAM VALIDATE-ID PERSONNEL-ID = &

END-HISTOGRAM

IF *NUMBER(H1.) = 0

  REINPUT 'Personnel ID does not exist, please retry. '

    MARK FIELD *&

END-IF

*

* SOLUTION 5.1.B - VERIFY SEX

*

IF & = 'F' OR = 'M' OR = ' ' 

  IGNORE

ELSE REINPUT 'Only “F”, “M”, or Blank is valid for Sex.'

  MARK FIELD *&

END-IF

*

* SOLUTION 5.1.D - VERIFY COUNTRY

*

IF & = 'USA' OR 'UK' OR = 'D'

  OR = 'DK' OR = 'F' OR = 'E'

    IGNORE

ELSE REINPUT

  'Only USA, UK, D, DK, F, or E are valid Country Codes.'

    MARK FIELD *&

END-IF

*

* SOLUTION 5.1.E - VERIFY PERSONNEL-ID TO COUNTRY

*

DECIDE FOR FIRST CONDITION

  WHEN PERSONNEL-ID = '10000000' THRU '19999999'

    AND COUNTRY = 'D'

      IGNORE

  WHEN PERSONNEL-ID = '20000000' THRU '29999999'

    AND COUNTRY = 'USA'

      IGNORE

  WHEN PERSONNEL-ID = '30000000' THRU '39999999'

    AND COUNTRY = 'UK'

      IGNORE

  WHEN PERSONNEL-ID = '40000000' THRU '49999999'

    AND COUNTRY = 'F'

      IGNORE

  WHEN PERSONNEL-ID = '50000000' THRU '59999999'

    AND COUNTRY = 'DK'

      IGNORE

  WHEN PERSONNEL-ID = '60000000' THRU '69999999'

    AND COUNTRY = 'E'

      IGNORE

  WHEN NONE

    REINPUT 'ID range invalid for Country Code.'

      MARK FIELD *&

END-DECIDE

*

* SOLUTION 5.1.F - PF-KEY PROCESSING RULE

*                  IMMEDIATE EXIT ROUTINE

IF & = ‘PF3

  ESCAPE ROUTINE

END-IF

*

* SOLUTION 5.2 - INPUT WINDOW

*

IF & = 'USA' OR 'UK' OR = 'D'

  OR = 'DK' OR = 'F' OR = 'E'

    IGNORE

ELSE

  SET CONTROL 'WFC11L30B14/45'

  INPUT WITH TEXT '    ** Country Code **'

        'Only the following codes'

      / 'are correct: '

      / 'USA...United States'

      / 'D.....Germany'

      / 'F.....France'

      / 'E.....Spain'

      / 'UK....United Kingdom'

      / 'DK....Denmark'

  SET CONTROL 'WB'

END-IF

Chapter 6 – Generating Help Objects

*

* EXCHLPR1 

*

DEFINE DATA PARAMETER 

1 #COUNTRY(A3) 

END-DEFINE 

DEFINE WINDOW COUNTRY 

  SIZE 4*60 

  BASE 17/10 

  TITLE 'Type a Country Code and press ENTER' 

  CONTROL SCREEN 

  FRAMED ON POSITION OFF 

INPUT WINDOW='COUNTRY' 

    'Only USA, UK, F, E, D, or DK are valid for Country. ' 

  / 'Enter a new value, please >' #COUNTRY (AD=MI'. ') 

IF #COUNTRY = 'USA' OR = 'UK' OR = 'F' 

      OR = 'E' OR = 'D' OR = 'DK' 

    IGNORE 

  ELSE REINPUT 'Please re-enter a valid Country Code' 

END-IF 

END

*

* EXCHLPR2 

*

DEFINE DATA PARAMETER 

1 #PERSONNEL-ID(A8) 

1 #ADDRESS(A20) 

LOCAL 

1 STAFF VIEW OF EMPLOYEES 

  2 PERSONNEL-ID 

  2 ADDRESS-LINE(1:3) 

1 #SELECT(A1/3) 

1 #INDEX(I2) 

END-DEFINE 

DEFINE WINDOW ADDRESS 

  SIZE 4*60 

  BASE 17/10 

  TITLE 'Mark your choice and press ENTER' 

  CONTROL SCREEN 

  FRAMED ON POSITION OFF 

FIND STAFF WITH PERSONNEL-ID = #PERSONNEL-ID 

  INPUT WINDOW='ADDRESS' 

    #SELECT(1) STAFF.ADDRESS-LINE(1) 

    #SELECT(2) STAFF.ADDRESS-LINE(2) 

    #SELECT(3) STAFF.ADDRESS-LINE(3) 

  EXAMINE #SELECT(*) FOR 'X' GIVING INDEX #INDEX 

  DECIDE FOR FIRST CONDITION 

    WHEN #INDEX = 1 

      #ADDRESS := STAFF.ADDRESS-LINE(#INDEX) 

    WHEN NONE 

      REINPUT 'Please make a single selection only.' 

  END-DECIDE 

END-FIND
END

*

* EXCH002A 

*

DEFINE DATA PARAMETER 

1 #JOB(A25) 

LOCAL 

1 H-VIEW VIEW OF EMPLOYEES 

  2 JOB-TITLE 

END-DEFINE 

H1. HISTOGRAM H-VIEW JOB-TITLE = #JOB 

END-HISTOGRAM 

IF *NUMBER(H1.) = 0 

  SET CONTROL 'WFC8L40B15/15' 

  INPUT WITH TEXT 'Job Title invalid as entered. ' 

      'Please enter a new value below' 

    / 'or press <ENTER> to continue: ' 

    / '>' #JOB 

  SET CONTROL 'WB' 

END-IF 

END 

*

* EXCH003A 

*

DEFINE DATA PARAMETER 

1 #CITY(A20) 

LOCAL 

1 H-VIEW VIEW OF EMPLOYEES 

  2 CITY 

END-DEFINE 

H1. HISTOGRAM H-VIEW CITY = #CITY 

END-HISTOGRAM 

IF *NUMBER(H1.) = 0 

  SET CONTROL 'WFC8L40B15/15' 

  INPUT WITH TEXT 'City invalid as entered. ' 

      'Please enter a new value below' 

    / 'or press <ENTER> to continue: ' 

    / '>' #CITY 

  SET CONTROL 'WB' 

END-IF 

END 

*

* EXCH004A 

*

DEFINE DATA PARAMETER 

1 #PID(A8) 

LOCAL 

1 H-VIEW VIEW OF EMPLOYEES 

  2 PERSONNEL-ID 

END-DEFINE 

SET CONTROL 'WFC8L40B15/15' 

IF #PID = ' ' 

  INPUT WITH TEXT 'Personnel ID required.' 

      'Please enter a value below' 

    / 'or press <ENTER> to continue:' 

    / '>' #PID 

  SET CONTROL 'WB' 

END-IF 

H1. HISTOGRAM H-VIEW PERSONNEL-ID = #PID 

END-HISTOGRAM 

IF *NUMBER(H1.) = 0 

  INPUT WITH TEXT 'Personnel ID invalid as entered.' 

      'Please enter a new value below' 

    / 'or press <ENTER> to continue:' 

    / '>' #PID 

  SET CONTROL 'WB' 

END-IF 

END 

Chapter 7 – Extended Help

0010 *

0020 * Helproutine & map to display JOB for selection

0030 *

0040 DEFINE DATA

0050 PARAMETER

0060 1 #JOB(A25)

0070 LOCAL

0080 1 EMPLOYEES-HIST VIEW OF EMPLOYEES

0090   2 JOB-TITLE

0100 1 #SELECT(A01/75)

0110 1 #VALUES(A25/75)

0120 1 #NUMBER(I2)

0130 1 #INDEX(I2)

0140 END-DEFINE

0150 SET KEY

0160   PF7='%W-'

0170   PF8='%W+'

0180 *

0190 * LOAD ARRAY VALUES

0200 *

0210 HISTOGRAM (75) EMPLOYEES-HIST JOB-TITLE

0220   ASSIGN #INDEX = *COUNTER

0230   MOVE JOB-TITLE TO #VALUES(#INDEX)

0240 END-HISTOGRAM

0250 SET CONTROL 'WFC15L30'

0260 *

0270 * PROCESS THE MAP

0280 *

0290 INPUT USING MAP 'EXCHLP02'

0300 EXAMINE #SELECT(*) FOR 'X'

0310   GIVING NUMBER #NUMBER INDEX #INDEX

0320 IF #NUMBER = 1

0330   MOVE #VALUES(#INDEX) TO #JOB

0340 ELSE

0350   REINPUT 'SELECT ONLY ONE ENTRY PLEASE'

0360 END-IF

0370 SET CONTROL 'WB'

0380 ESCAPE ROUTINE

0390 END

0010 *

0020 * Helproutine invokes map to display JOB TITLE for selection

0030 *

0040 DEFINE DATA

0050 PARAMETER

0060 1 #JOB(A25)

0070 LOCAL

0080 1 EMPLOYEES-HIST VIEW OF EMPLOYEES

0090   2 JOB-TITLE

0100 1 #SELECT(A01/75)

0110 1 #VALUES(A25/75)

0120 1 #NUMBER(I2)

0130 1 #INDEX(I2)

0140 END-DEFINE

0150 SET KEY

0160 PF7='%W-'

0170 PF8='%W+'

0180 *

0190 * LOAD ARRAY VALUES

0200 *

0210 HISTOGRAM (75) EMPLOYEES-HIST JOB-TITLE

0220   ASSIGN #INDEX = *COUNTER

0230   MOVE JOB-TITLE TO #VALUES(#INDEX)

0240 END-HISTOGRAM

0250 DEFINE WINDOW GETAJOB

0260   SIZE 15*30

0270   BASE CURSOR

0280   TITLE 'Type "X" to select a job'

0290   CONTROL SCREEN

0300   FRAMED ON POSITION OFF

0310 *

0320 * PROCESS THE MAP

0330 *

0340 INPUT WINDOW='GETAJOB' USING MAP 'EXCHLP02'

0350 EXAMINE #SELECT(*) FOR 'X'

0360   GIVING NUMBER #NUMBER INDEX #INDEX

0370 IF #NUMBER = 1

0380   MOVE #VALUES(#INDEX) TO #JOB

0390 ELSE

0400   REINPUT 'SELECT ONLY ONE ENTRY PLEASE'

0410 END-IF

0420 ESCAPE ROUTINE

0430 END

Chapter 8 – DEFINE WINDOW

DEFINE DATA LOCAL

1 #FROM   (A08)

1 #THRU   (A08)

1 #COM    (A40)

1 #TEXT   (A70) INIT <'You must enter a starting value,'

                   -  'a direct command, or quit this'

                   -  'function.'>

END-DEFINE

INCLUDE EXCCKEYS    /* Copycode for PF key lines

DEFINE WINDOW CTLSCR

  SIZE 8 * 35

  BASE 8 / 8

  TITLE ' State Your Action '

  FRAMED ON

REPEAT

  INPUT USING MAP 'EXCTESTM'

  DECIDE FOR FIRST CONDITION

    WHEN *COM = ' '

    AND #FROM = ' '

   INPUT WINDOW='CTLSCR' WITH TEXT #TEXT

        / 'A starting value is required.' (I)

        / 'Type a starting value or type'

        / 'a period to quit.'

        / '>>' #FROM (AD=MI'_')

      ESCAPE TOP

    WHEN *COM GT ' '

      #COM := *COM

      PROCESS COMMAND-VIEW

        ACTION EXEC USING PROCESSOR-NAME='EXCCP'

        COMMAND-LINE(1) = #COM

    WHEN #BEGIN GT ' '

      FETCH 'EXCREV01' #FROM #THRU

    WHEN NONE IGNORE

  END-DECIDE

END-REPEAT

END

* SOLUTION 5.2 - INPUT WINDOW

* MODIFIED FOR SOLUTION 8.1.A

*

DEFINE WINDOW COUNTRY

  SIZE 10 * 30

  BASE BOTTOM RIGHT

  TITLE '** Country Code **'

  CONTROL SCREEN

  FRAMED ON POSITION OFF

IF & = 'USA' OR 'UK' OR = 'D'

  OR = 'DK' OR = 'F' OR = 'E'

    IGNORE

ELSE

INPUT WINDOW='COUNTRY'

        'Only the following codes'

      / 'are correct: '

      / 'USA...United States'

      / 'D.....Germany'

      / 'F.....France'

      / 'E.....Spain'

      / 'UK....United Kingdom'

      / 'DK....Denmark'

END-IF

*

* EXCH002A

* MODIFIED FOR SOLUTION 8.1.B

*

DEFINE DATA PARAMETER

1 #JOB(A25)

LOCAL

1 H-VIEW VIEW OF EMPLOYEES

  2 JOB-TITLE

END-DEFINE

DEFINE WINDOW JOBS

  SIZE 10 * 30

  BASE BOTTOM RIGHT

  TITLE 'Job Invalid as Entered'

  CONTROL SCREEN

  FRAMED ON POSITION OFF

H1. HISTOGRAM H-VIEW JOB-TITLE = #JOB

END-HISTOGRAM

IF *NUMBER(H1.) = 0

  INPUT WINDOW='JOBS'

      'Please enter a new value'

    / 'below or press <ENTER> to'

    / 'continue:'

    / '>' #JOB

END-IF

END

*

* EXCH003A

* MODIFIED FOR SOLUTION 8.1.C

*

DEFINE DATA PARAMETER

1 #CITY(A15)

LOCAL

1 H-VIEW VIEW OF EMPLOYEES

  2 CITY

END-DEFINE

DEFINE WINDOW CITY

  SIZE 10 * 30

  BASE BOTTOM RIGHT

  TITLE 'CITY Invalid as Entered'

  CONTROL SCREEN

  FRAMED ON POSITION OFF

H1. HISTOGRAM H-VIEW CITY = #CITY

END-HISTOGRAM

IF *NUMBER(H1.) = 0

  INPUT WINDOW='CITY'

      'Please enter a new value'

    / 'below or press <ENTER> to' 

    / 'continue:'

    / '>' #CITY

END-IF

END

0010 *

0020 * Helproutine invokes map to display MAKE for selection

0030 *

0040 DEFINE DATA

0050 PARAMETER

0060 1 #JOB(A25)

0070 LOCAL

0080 1 EMPLOYEES-HIST VIEW OF EMPLOYEES

0090   2 JOB-TITLE

0100 1 #SELECT(A01/75)

0110 1 #VALUES(A25/75)

0120 1 #NUMBER(I2)

0130 1 #INDEX(I2)

0140 END-DEFINE

0150 DEFINE WINDOW JOBS

0160   SIZE 10 * 30

0170   BASE BOTTOM RIGHT

0180   TITLE 'Mark with "X" to Select'

0190   CONTROL SCREEN

0200   FRAMED ON POSITION OFF

0210 SET KEY

0220   PF7='%W-'

0230   PF8='%W+'

0240 *

0250 * LOAD ARRAY VALUES

0260 *

0270 HISTOGRAM (75) EMPLOYEES-HIST JOB-TITLE

0280   ASSIGN #INDEX = *COUNTER

0290   MOVE JOB-TITLE TO #VALUES(#INDEX)

0300 END-HISTOGRAM

0310 *

0320 * PROCESS THE MAP

0330 *

0340 INPUT WINDOW='JOBS' USING MAP 'EXCHLP02'

0350 EXAMINE #SELECT(*) FOR 'X'

0360   GIVING NUMBER #NUMBER INDEX #INDEX

0370 IF #NUMBER = 1

0380   MOVE #VALUES(#INDEX) TO #JOB

0390 ELSE

0400   REINPUT 'SELECT ONLY ONE ENTRY PLEASE'

0410 END-IF

0420 ESCAPE ROUTINE

0430 END

0010 DEFINE DATA LOCAL

0020 1 CARS VIEW OF VEHICLES

0030  2 PERSONNEL-ID

0040  2 MAKE

0050  2 MODEL

0060 END-DEFINE

0070 INPUT 'ENTER A MAKE:' CARS.MAKE (HE='WHATCAR' AD=M)

0080 FIND CARS WITH MAKE = CARS.MAKE

0090   IF NO RECORDS FOUND

0100     REINPUT USING HELP MARK *CARS.MAKE

0110   END-NOREC

0120   DISPLAY PERSONNEL-ID MAKE MODEL

0130 END-FIND

0140 END

0010 * HELPROUTINE: WHATCAR

0020 DEFINE DATA PARAMETER

0030 1 #MAKE(A20)

0040 LOCAL

0050 1 HCARS VIEW OF VEHICLES

0060  2 MAKE

0070 END-DEFINE

0080 FORMAT PS=15

0090 DEFINE WINDOW CARS

0100   SIZE 18 * 35

0110   BASE 1 / 35

0120   TITLE 'Type Your Selection'

0130   CONTROL SCREEN

0140   FRAMED ON POSITION OFF

0150 SET WINDOW='CARS'

0160 HISTOGRAM HCARS MAKE

0170   PRINT NOTITLE MAKE *NUMBER(AD=L)

0180   AT END OF PAGE

0190     INPUT 'ENTER MAKE:' #MAKE

0200     IF #MAKE NE ' ' 

0210       ESCAPE ROUTINE

0220     END-IF

0230   END-ENDPAGE

0240   AT END OF DATA

0250     ESCAPE ROUTINE

0260   END-ENDDATA

0270 END-HISTOGRAM

0280 END

0310 *


0320 * PROCESS THE MAP


0330 *


0340 INPUT WINDOW='GETAJOB' USING MAP 'EXCHLPR3'


0350 EXAMINE #SELECT(*) FOR 'X'


0360   GIVING NUMBER #NUMBER INDEX #INDEX


0370 IF #NUMBER = 1


0380   MOVE #VALUES(#INDEX) TO #JOB


0390 ELSE


0400   REINPUT 'SELECT ONLY ONE ENTRY PLEASE'


0410 END-IF


0420 ESCAPE ROUTINE


0430 END


0340 FOR #INDEX 1 75


0350   IF *CURS-FIELD = POS(#VALUES(#INDEX))


0360     SET CONTROL 'CCC1'


0370     #JOB := *COM


0380     ESCAPE ROUTINE


0390   END-IF


0400 END-FOR


0410 ESCAPE ROUTINE


0420 END

*

* EXCPGM9A

*

DEFINE DATA LOCAL

1 #CLINE(A75) INIT <'CREATE  MODIFY  UPDATE  LIST  MAIN  QUIT'>

1 #COM(A20)

END-DEFINE

SET KEY PF5

PF7='%W-'

PF8='%W+'

DEFINE WINDOW PULLDOWN

  CONTROL SCREEN

  FRAMED ON POSITION OFF

SET CONTROL 'M21'

INPUT (AD=O IP=OFF) #CLINE

IF *CURS-LINE = 1

  SET CONTROL 'CCCW'

  #COM := *COM

  DECIDE ON FIRST VALUE OF #COM

    VALUE 'CREATE' IGNORE

    VALUE 'MODIFY' IGNORE

    VALUE 'UPDATE'

      INPUT WINDOW='PULLDOWN'

        'VEHICLES' / 'EMPLOYEES' / 'EXIT'

    VALUE 'LIST'

       INPUT WINDOW='PULLDOWN' USING MAP 'EXCHTXT1'

       IF *PF-KEY = 'PF5'

         SET CONTROL 'CCCW'

         #COM := *COM

         DECIDE ON FIRST VALUE OF #COM

           VALUE 'Exit' ESCAPE BOTTOM

           VALUE 'Vehicles' IGNORE

           VALUE 'Employees'

             CALLNAT 'EXCN001' #COM

             PRINT #COM

           NONE IGNORE

         END-DECIDE

       END-IF

    VALUE 'MAIN' IGNORE

    VALUE 'QUIT' ESCAPE BOTTOM

    NONE IGNORE

  END-DECIDE

END-IF

END

*

* EXCN001 - Subprogram processing EMPLOYEES selection

*

DEFINE DATA

PARAMETER

1 #COM(A20)

END-DEFINE

SET KEY PF5

  PF7='%W-'

  PF8='%W+'

DEFINE WINDOW XVALUES

  SIZE 8 * 16

  TITLE '- Employees -'

  CONTROL SCREEN

  FRAMED ON

INPUT WINDOW='XVALUES'

  TEXT 'Place the cursor on selection and press PF5'

  USING MAP 'EXCHTXT2'

IF *PF-KEY = 'PF5'

  SET CONTROL 'CCC1' /* Use 1 here because of Job Title

  #COM := *COM

  CALLNAT 'EXCN002' #COM

END-IF

END

*

* EXCN002 - Subprogram processing EMPLOYEES listing

*

DEFINE DATA

PARAMETER

1 #COM(A20)

LOCAL

1 H-VIEW-NAME VIEW OF EMPLOYEES       /* Multiple views set up

  2 NAME                              /* for HISTOGRAMS; one for

1 H-VIEW-CITY VIEW OF EMPLOYEES       /* each field in the list

  2 CITY                              /* presented in EXCHTXT2

1 H-VIEW-JOB  VIEW OF EMPLOYEES

  2 JOB-TITLE

1 #VALUES(A20/100)

1 #COUNTER(I2)

END-DEFINE

SET KEY PF5

  PF7='%W-'

  PF8='%W+'

DEFINE WINDOW XVALUES

  SIZE 12 * 25

  TITLE '- Values -'

  CONTROL SCREEN

  FRAMED ON

*

DECIDE FOR EVERY CONDITION     /* This section detects which

  WHEN #COM GT ' '             /* selected field will be

    IF #COM = SCAN 'Name'      /* processed.

      PERFORM LOAD-NAME        /* Subroutine that loads map

    END-IF                     /* array for selection.

  WHEN #COM GT ' '

    IF #COM = SCAN 'City'

      PERFORM LOAD-CITY

    END-IF

  WHEN #COM GT ' '

    IF #COM = SCAN 'Job Title'

      PERFORM LOAD-JOB

    END-IF

  WHEN NONE IGNORE

END-DECIDE

*

INPUT WINDOW='XVALUES'

  TEXT 'Place the cursor on your selection and press PF5.'

  USING MAP 'EXCHTXT3'

IF *PF-KEY = 'PF5'

  SET CONTROL 'CCC1'            /* The number 1 indicates a

  #COM := *COM                  /* field of data is to be

  ESCAPE MODULE                 /* copied not just the word

END-IF                          /* that the W indicates. This

*                               /* is needed because of any

* Load NAME Array               /* element with an embedded

*                               /* space(s).

DEFINE LOAD-NAME

HISTOGRAM (100) H-VIEW-NAME NAME

  #COUNTER := *COUNTER

  MOVE NAME TO #VALUES(#COUNTER)

END-HISTOGRAM

END-SUBROUTINE

*

* Load CITY Array

*

DEFINE LOAD-CITY

HISTOGRAM (100) H-VIEW-CITY CITY

  #COUNTER := *COUNTER

  MOVE CITY TO #VALUES(#COUNTER)

END-HISTOGRAM

END-SUBROUTINE

*

* Load JOB-TITLE Array

*

DEFINE LOAD-JOB

HISTOGRAM (100) H-VIEW-JOB JOB-TITLE

  #COUNTER := *COUNTER

  MOVE JOB-TITLE TO #VALUES(#COUNTER)

END-HISTOGRAM

END-SUBROUTINE

*

END

*

* SOLUTION 9.1

*

DEFINE DATA LOCAL

1 STAFF VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 NAME

  2 FIRST-NAME

  2 MIDDLE-I

  2 SEX

  2 BIRTH

  2 ADDRESS-LINE(3)

  2 CITY

  2 COUNTRY

  2 ZIP

  2 JOB-TITLE

  2 DEPT

  2 SALARY(1)

1 CARS VIEW OF VEHICLES

  2 PERSONNEL-ID

  2 MAKE

  2 MODEL

  2 COLOR

  2 YEAR

  2 CLASS

1 #COM(A20)

END-DEFINE

SET KEY ALL

SET KEY PF1 = HELP

  PF3 NAMED 'EXIT'

  PF7='%W-' NAMED 'BKWD'

  PF8='%W+' NAMED 'FWD'

DEFINE WINDOW PORTHOLE

  CONTROL SCREEN

  FRAMED ON

SET CONTROL 'M21'

READ (5) EMPLOYEES BY PERSONNEL-ID

    STARTING FROM '20000100'

  INPUT USING MAP 'EXCMAP01'

  IF #CURS-LINE = 1

    SET CONTROL 'CCCW'

    #COM := *COM

    DECIDE ON FIRST VALUE OF #COM

      VALUE 'STAFF-LIST'

        FETCH RETURN 'EXCP015A'

      VALUE 'CITY-LIST'

        FETCH RETURN 'EXCP014A'

      VALUE 'JOB-LIST'

        FETCH RETURN 'EXCP013A'

      VALUE 'CAR-LIST'

        CALLNAT 'EXCN015A' STAFF.PERSONNEL-ID

      NONE IGNORE

    END-DECIDE

  END-IF

END-READ

END

*

* EXCP015A

*

DEFINE DATA LOCAL

1 STAFF VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 NAME

  2 FIRST-NAME

  2 MIDDLE-I

  2 SEX

  2 BIRTH

END-DEFINE

READ STAFF BY NAME

DISPLAY PERSONNEL-ID NAME FIRST-NAME MIDDLE-I SEX BIRTH

END

*

* EXCP014A

*

DEFINE DATA LOCAL

1 STAFF VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 NAME

  2 FIRST-NAME

  2 MIDDLE-I

  2 CITY

END-DEFINE

READ STAFF BY CITY

PRINT PERSONNEL-ID NAME FIRST-NAME MIDDLE-I 50T CITY

END-READ

END

*

* EXCP013A

*

DEFINE DATA LOCAL

1 STAFF VIEW OF EMPLOYEES

  2 PERSONNEL-ID

  2 NAME

  2 FIRST-NAME

  2 MIDDLE-I

  2 JOB-TITLE

READ STAFF BY JOB-TITLE

PRINT PERSONNEL-ID NAME FIRST-NAME MIDDLE-I 50T JOB-TITLE

END-READ

END

*

* EXCN015A

*

DEFINE DATA

PARAMETER

1 #ID(A8)

LOCAL

1 CARS VIEW OF VEHICLES

  2 PERSONNEL-ID

  2 MAKE

  2 MODEL

  2 COLOR

  2 YEAR

  2 CLASS

1 #MAKE(A30/5)

1 #MARK(A1/5)

1 #CV(C/5)

1 #ISN(P10/5)

1 #COUNTER(I2)

END-DEFINE

*

* Window for general display of CARS

*

DEFINE WINDOW AUTOS

  TITLE ‘- MARK TO SELECT DETAIL -

  CONTROL SCREEN

  FRAMED ON

*

* Window for detail display of CARS

*

DEFINE WINDOW MORECAR

  TITLE ‘- DETAIL OF VEHICLE -

  CONTROL SCREEN

  FRAMED ON

*

* FIND executed for passed PERSONNEL-ID

*

FIND (5) CARS WITH PERSONNEL-ID = #ID

*

* If this person doesn’t own an automobile

*

  IF NO RECORDS FOUND

    INPUT WINDOW=‘AUTOS’ ‘NO AUTOS ON RECORD

    ESCAPE ROUTINE

  END-NOREC

*

* Load up the array for display

*

  #COUNTER := *COUNTER

  COMPRESS MAKE MODEL INTO #MAKE(#COUNTER)

  #ISN(#COUNTER) = *ISN

END-FIND

*

* If the array doesn’t fill up, we don’t want the user

* to be able to ask for the detail of a blank line.

* This code temporarily protects all #MARK fields.

*

ADD 1 TO #COUNTER

FOR #INDEX #COUNTER 5

  MOVE (AD=P) TO #CV(#INDEX)

END-FOR

INPUT WINDOW=‘AUTOS

  #MARK(1) (AD=MI’_’ CV=#CV(1)) #MAKE(1) (AD=O) /

  #MARK(2) (AD=MI’_’ CV=#CV(2)) #MAKE(2) (AD=O) /

  #MARK(3) (AD=MI’_’ CV=#CV(3)) #MAKE(3) (AD=O) /

  #MARK(4) (AD=MI’_’ CV=#CV(4)) #MAKE(4) (AD=O) /

  #MARK(5) (AD=MI’_’ CV=#CV(5)) #MAKE(5) (AD=O) /

*

* Check the selection array for a choice and process

* only the first selection requested.

*

  EXAMINE #MARK(*) FOR ‘X’ GIVING INDEX #INDEX

  IF #INDEX GT 0

*

* Retrieve the car by ISN

*

    GET CARS #ISN(#INDEX)

    INPUT WINDOWS=‘MORECAR

      MAKE(G1.) / MODEL(G1.) /

      COLOR(G1.) / YEAR(G1.) / CLASS(G1.)

  END-IF

END

0010 *                              

0020 * EXCPGM03                     

0030 *                              

0040 DEFINE DATA                    

0050 LOCAL USING EXCCMD             

0060 LOCAL                          

0070 1 EMPLOYEES VIEW OF EMPLOYEES  

0080   2 PERSONNEL-ID               

0090   2 NAME                       

0100   2 FIRST-NAME                 

0110   2 MIDDLE-I                   

0120   2 SEX                        

0130   2 BIRTH                      

0140   2 ADDRESS-LINE(1)            

0150   2 CITY                       

0160   2 COUNTRY                    

0170   2 POST-CODE                  

0180   2 DEPT                       

0190   2 JOB-TITLE                  

0200   2 SALARY(1)                  

0210 1 #PROGRAM(A8) INIT <*PROGRAM>                      

0220 1 #COM(A60)                                         

0230 1 #PERSONNEL-ID(A8)                                 

0240 END-DEFINE                                          

0250 REPEAT                                              

0260   INPUT USING MAP 'EXCMAP00'                        

0270   IF #PERSONNEL-ID = SCAN '.'                       

0280     ESCAPE MODULE                                   

0290   END-IF                                            

0300   FIND EMPLOYEES WITH PERSONNEL-ID = #PERSONNEL-ID  

0310     INPUT USING MAP 'EXCMAP03'                      

0320     IF #COM GT ' '                                  

0330       PROCESS COMMAND-VIEW ACTION                   

0340         EXEC USING PROCESSOR-NAME='EXCCP'           

0350         COMMAND-LINE(1) = #COM                      

0360     END-IF                                          

0370   END-FIND                                          

0380 END-REPEAT                                          

0390 END                              

